
Lucrarea nr.4

Mediul de simulare SPICE


1. SCOPUL LUCRĂRII
Se urmăreşte prezentarea unor noţiuni elementare de sintaxă, modelare, analiză şi simulare pentru circuite analogice, oferite de programele din familia SPICE.

2. InTroducere


Conceput încă din anii ’70 la University of California, Berkeley, Spice (Simulation Program with Integrated Circuit Emphazis) este un program de simulare de uz general  pentru circuitelor electronice, analogice şi digitale. În scurt tip el a devenit un standard industrial, datorită utlităţii, funcţionalităţii sale dar şi datorită distribuţiei libere a programelor sursă (SPICE2G6, SPICE3f4). Versiunile comerciale existente sub diferite denumiri (ex.: MicroSim Pspice v.8, OrCAD v.9.2) au ajuns să ofere solutii EDA (Electronic Design Automation) complete; mediul integrat permite parcurgerea tuturor etapelor (proiectare, desenare, simulare, optimizare, proiectare cablaj imprimat) aferente realizării unui produs electronic în mod automat.

Avantajele pe care le oferă acest mediu sunt:

· reducerea cheltuielilor de proiectare; simularea unui circuit este mult mai ieftină decât experimentarea aceluiaşi circuit în laborator;

· dezvolatarea rapidă a prototipurilor;

· analize multiple; SPICE poate să efectueze toate analizele care se fac în mod tradiţional într-un laborator dar şi cele inaccesibile în mod normal (simulari ale variaţiei temperaturii, verificarea funcţionării unei arii dintr-un circuit integrat fizic inaccesibil proiectantului etc.).

Pentru a descrie topologia unui circuit într-un mod acceptat de simulatorul Spice, este nevoie de un fişier tip listă de legături (netlist). Acesta poate fi generat automat, pe baza schemei desenate într-un editor specializat, sau poate fi creat manual cu ajutorul oricărui editor de text.



În cele ce urmează vor fi detaliate aspecte legate de definirea circuitelor, discuţia orientându-se doar asupra componentelor cel mai des folosite în circuitele analogice.

Elementele de sintaxă au fost preluate după specificaţiile produselor PSpice v5…v8 al firmei MicroSim Corporation.


Observaţie: În cele ce urmează notaţia “<.>” va desemna un câmp obligatoriu iar “[.]” un câmp opţional. Notaţia “<.>*” specifică posibilitatea existenţiei subsecvente a mai multori câmpuri de acelaşi tip.
3. COMPONENTE ŞI DECLARAŢII

3.1 Componente pasive


Dintre componentele pasive disponibile în SPICE se pot enumera: rezistor, condensator, bobină, transformator, linie de transmisie. Acestea încep respectiv cu literele R, C, L, K şi T. Rezistoarele pot avea şi coeficienţi de temperatură. Declaraţiile prin care aceste componente se specifică în circuit sunt următoarele:

· R<nume> <nod+> <nod-> [<model>] <valoare>

Exemplu: R1 15 0 2K;

· C<nume> <nod+> <nod-> [<model>] <valoare> [IC=<valoare iniţială>] 

Exemplu: CCUPLAJ 15 0 10uF; C1 3 33 CMODEL 10pF IC=1.5v;

· L<nume> <nod+> <nod-> [model] <valoare> [IC=<initial>] 

Exemplu: LSARCINA 15 0 20mH; L2 5 12 2uH IC=2mA;

· K<nume> L<nume inductanţă> <L< nume inductanţă>> + <coeficient de cuplaj> 

Exemplu: KXFR1   LPRIM LSEC  .99;

· T<nume> <A+> <A-> <B+> <B-> Z0=<valoare>

+ [TD=<valoare> | F=<valoare>[NL=<valoare>]] 

Exemplu: T1   1  2  3  4  Z0=220  TD=115ns.


3.2 DISPOZITIVE SEMICONDUCTOARE
Dintre dispozitivele semiconductoare disponibile în SPICE se pot enumera: diode, tranzistoare bipolare, tranzistoare cu efect de câmp: cu joncţiune, MOS, cu GaAs. Acestea încep respectiv cu literele D, Q, J, M, B. Toate aceste componente necesită modele specificate de utilizator sau existente biblioteci. Declaraţiile prin care aceste componente se specifică în circuit sunt următoarele:

· D<nume> <nod+> <nod-> <model>

Exemplu: D1 2 3 D1N4148

· Q<nume> < c > < b > < e > + [<nod substrat>] <model>

Exemplu: Q1 14 2 13  BC107

· J<nume> < d > < g > < s > <model>

Exemplu: J1 20  1  0  J2N3819 

· M<nume> < d > < g > < s > +<substrat> <model>

Exemplu: M1 1 2 0 0 IRF150

· B<nume> < d > < g > < s > <model>

Exemplu: BIN 100  1  0  GFAST.

3.3 Surse de curent şi tensiune


Acestea sunt singurele componente generatoare de putere. Sursele de curent pot fi comandate sau independente.

a) Surse comandate. În SPICE sunt disponibile patru tipuri de surse comandate: surse de tensiune comandate prin tensiune, surse de curent comandate prin curent, surse de curent comandate prin tensiune, surse de tensiune comandate prin curent.

Acestea încep respectiv cu literele E, F, G, H implementând funcţiile:

va = E(vb)
ia = F(ib)
I a= G(vb)
va = H(ib)

Sursele pot fi comandate prin funcţii liniare sau polinomiale de tipul:

 O funcţie polinomială bidimensională este exprimată astfel:

 Declaraţiile corespunzătoare au următoarea formă generală:
· E<nume> <nod(+)> <nod(-)> <nod_comandă(+)> <nod_comandă(-)> <amplificare>

· E<nume> <nod(+)> <nod(-)> POLY(<valoare>) <nod_comandă(+)> <nod_comandă(-)> <<valoare (coeficient polinomial)>>*
Exemple:

E1 1 2 10 11 48.21 – specifică o sursă de tensiune comandată prin tensiune între nodurile 1 şi 2, liniară, comandată de tensiunea dintre nodurile 10 şi 11: VE1_1-2 = 48,21*V10-11
ENELIN 1 2 POLY(2) 3 4 5 6 0.1 13.6 0.2 0.005 – specifică o sursă de tensiune comandată prin tensiune între nodurile 1 şi 2, neliniară, comandată de tensiunile dintre nodurile 3 şi 4 respectiv 5 şi 6: VNELIN1-2 = 0,1 + 13,6*V3-4 + 0.2*V5-6 + 0.005*V23-4
· F<nume> <nod(+)> <nod(-)> <nume (dispozitiv comandă tensiune) <amplificare>
· F<nume> <nod(+)> <nod(-)> POLY(<valoare>) <nume (dispozitiv comandă tensiune) <<valoare (coeficient polinomial)>>*

Exemple:

FSENS 1 2 VSENS 10.0 – specifică o sursă de curent liniară, între nodurile 1 şi 2, comandată de curentul ce circulă prin sursa de tensiune VSENS: IFSENS = 10,0*I(VSENS).

FNELIN 101 102 POLY(2) V1 V2 0.0 0.9 0.2 0 1 – specifică o sursă de curent neliniară, între nodurile 101 şi 102, comandată de curentul ce circulă prin sursele de tensiune V1 şi V2: IFNELIN = 0,9*I(V1) + 0,2*I(V2) +1* I(V1)* I(V2)
· G<nume> <nod(+)> <nod(-)> <<nod_comandă(+)> <nod_comandă(-)>> <transconductanţă>

· G<nume> <nod(+)> <nod(-)> POLY(<valoare>) <<nod_comandă(+)> <nod_comandă(-)>> <<valoare (coeficient polinomial)>>*

· H<nume> <nod(+)> <nod(-)> <nume (dispozitiv comandă tensiune) <transrezistenţă>

· H<nume> <nod(+)> <nod(-)> POLY(<valoare>) <nume (dispozitiv comandă tensiune) <<valoare (coeficient polinomial)>>*

b) Surse independente. Fiecare sursă poate fi specificată separat pentru analizele în curent continuu, în curent alternativ sau analiza în domeniul timp. Sursele independente de tensiune şi semnal încep cu litera V:

· V<nume> <nod+> <nod-> + [[DC] <valoare>] + [AC <amplitudine> [<faza>]] + [ <specificaţii de regim tranzitoriu> ]. 

Specificaţiile de regim tranzitoriu: EXP(iv vpk rdelay rtc fdelay ftc), PULSE(v1 v2 td trise tfall pw per), PWL(t1 v1 t2 v2 ... tn vn), SFFM(voff vampl fc mod fm),SIN(voff vampl freq td df phase).

Exemplu: V1 13 0 2.3mV; VAC 2 3  AC .001;

   V3 26 77 DC .002 AC 1 SIN(.002 .002 1.5MEG).

iar sursele independente de curent şi semnal cu I:
· I<nume> <nod+> <nod-> + [[DC] <valoare>] + [AC <amplitudine> [<faza>]] +

[ <specificaţii de regim tranzitoriu> ].

Specificaţiile de regim tranzitoriu: EXP(i1 ipk rdelay rtc fdelay ftc), PULSE(i1 i2 td trise tfall pw per), PWL(t1 i1 t2 i2 ... tn fn), SFFM(ioff iampl fc mod fm), SIN(ioff iampl freq td df phase).

Exemplu: IBIAS 13 0 2.3mA; I2 2 3 AC .001;

   I3 26 77 DC .002 AC 1 SIN(.002 .002 1.5K).
3.4 Comutatoare

Acestea sunt implementate pentru a permite schimbarea conexiunilor circuitelor în timpul simulării. Comutatoarele pot fi comandate prin curent sau tensiune; pentru a crea un comutator comandat comandat în timp se conectează nodurile de control al comutatorului la o sursă de tensiune cu valori de tensiune egale cu cele de timp.

Comutatorul nu este ideal pentru că rezistenţele de deschidere, respectiv de închidere sunt finite şi se modifică brusc între cele două valori.

În practică, rezistenţa de deschidere poate fi foarte mică în comparaţie cu impedanţele altor circuite, iar rezistenţa de închidere poate fi foarte mare în comparaţie cu impedanţa altor circuite.

Un comutator comandat prin tensiune se apelează cu declaraţia .MODEL de tipul VSWITCH, iar un comutator comandat prin curent cu declaraţia .MODEL de tipul ISWITCH. Forma generală a comutatorului comandat prin tensiune este precedată de litera S iar a comutatorului comandat prin curent W:

· S<nume> <nod+> <nod-> + <nod comandă+> <nod comandă-> <model>

Exemplu: S12 13 17 2 0 SMOD; SRESET 5 0 15 3 RELAY.

· W<nume> <nod+> <nod-> + <nume dispozitiv comandă tensiune> <model>

Exemplu: W12 13 17 VC WMOD ; WRESET 5 0 VRESET RELAY.

3.5 Alte declaraţii

a)  Decalaraţia .MODEL. Este utilizată pentru a defini un set de parametri ce descriu un dispozitiv şi are următoarea formă:

· .MODEL <nume model> <nume tip> +  [<nume parametru>=<valoare> [<tol>]]

Exemplu:  .MODEL Q1 NPN (IS=1e-7 BF=30).
b) Declaraţia .LIB. Bibliotecile sunt fişiere de text ASCII conţinând declaraţii .MODEL care definesc fiecare dispozitiv acceptat de SPICE. Toate fişierele bibliotecă SPICE au extensia .LIB.

Formatul comenzii .LIB este:

· .LIB [nume fişier]
Exemplu: .LIB F:\ELECTRO\DCE.LIB; SPICE va căuta în calea specificată fişierul DCE.LIB.

c) Declaraţia .PARAM. Cu ajutorul acestei declaraţii se pot defini parametrii globali ai circuitului:

· .PARAM < <nume>=<valoare> >* 

· .PARAM < <nume>={<expresie>}>* 

Exemplu: .PARAM   pi=3.14159265


d) Declaraţia .IC. Semnifică specificaţia condiţiilor iniţiale pentru punctul de polarizare:

· .IC < V(<nod>) = <valoare> >*

Exemplu: .IC V(2)=3.4  V(102)=0


e) Declaraţia .END. Ea marchează sfârşitul fişierului. Forma generală este:

· .END


4. ANALIZE

Aceste comenzi se plaseză în fişierul circuitului împreună cu definiţia circuitului care se doreşte a fi simulat. Toate comenzile sunt conţinute în declaraţii care încep cu “.”

4.1 aNALIZA în curent continuu
Se presupune că se baleiază o sursă de tensiune sau de curent, un parametru al modelului sau temperatura într-un anumit domeniu. Calculul punctului de polarizare se face pentru fiecare valoare baleiată. Analiza se specifică prin comanda .DC:

· .DC [LIN] <nume variabilă> + <start> <sfârşit> <increment>

· .DC [OCT][DEC] <nume variabilă> + <start> <sfârşit> <număr de puncte>

· .DC <nume variabilă> LIST + <valoare>*

Exemplu:

.DC  VIN  -7.25 7 .25  .05; .DC  LIN I2 5mA -2mA 0.1mA

.DC  TEMP LIST 0 20 27 50 80; .DC  PARAM RS -1 1 0.1
4.2 Analiza în curent alternativ (în DOMENIUL frecvenţă)

Analiza în curent alternativ calculează răspunsul circuitului în complex, la semnale mici, acestea fiind calculate prin liniarizarea circuitului în jurul punctului de funcţionare. Analiza se specifică prin comanda .AC:

· .AC [LIN][OCT][DEC] <nr. de pucte> <start> <sfârşit>
Exemplu: .AC  LIN  101  10Hz  200Hz; .AC  OCT   10  1KHz  16KHz;

                .AC  DEC   20  1MEG 100MEG

4.3 ANALIZA ÎN DOMENIUL TIMP
Declaraţia .TRAN determină calculul răspunsului în regim tranzitoriu pornind de la TIME = 0 până la o valoare specificată. În timpul analizei, SPICE fixează un pas de timp intern care este tot timpul ajustat pentru a se menţine acurateţea calculului. În domenii cu variaţii nesemnificative, pasul intern este incremenat iar în timpul domeniilor cu variaţii semnificative este decrementat. Dacă se doresc parametrii de semnal mic ai punctului de polarizare trebuie utilizate declaraţiile .TRAN/.OP:

· .TRAN[/OP] <pas tipărire> <valoare timp final>

 + [<pas netipărite> [<limită>]] [UIC]

Exemplu: .TRAN 1nS 100nS; .TRAN/OP 1nS 100nS 20nS UIC 

  .TRAN 1nS 100nS 0nS .1nS

4.4 Analiza statistică Monte Carlo
În acest regim este necesar să se specifice toleranţele valorilor parametrilor modelului. Această analiză permite rularea multiplă a celorlalte analize, cu valori luate aleator în domeniul lor de toleranţe. Există două tipuri de toleranţe: DEV şi LOT. Toleranţele DEV sunt independente, în timp ce toleranţele LOT sunt dependente.

Odată stabilite toleranţele, analiza Monte Carlo se lansează cu .MC:

· .MC <#rulări> [DC][AC][TRAN] + <var.de ieşire> <funcţie> [opţiune]* 

Exemplu: .MC  10 TRAN V(5) YMAX; .MC 50 DC IC(Q7) MIN LIST 


4.5 Analiza comportării circuitelor cu temperatura


SPICE permite să se analizeze un circuit la orice temperatură. Valoarea arbitrară a temperaturii la care se face analiza este 27°C. Această valoare poate fi modificată prin directiva .TEMP:

· .TEMP <valoare>* 

Exemplu: .TEMP 125; .TEMP 0 27 100.


4.6 Analiza parametrică


SPICE poate să realizeze rulări multiple, variind temperatura sau diferiţi parametrii ai modelului. Se utilizează comanda .STEP:

· .STEP [LIN] <nume variabilă> +  <start> <sfârşit> <increment>

· .STEP [OCT][DEC] <nume variabilă> + <start> <sfârşit> <număr de puncte>

· .STEP <nume variabilă> LIST  + <valoare>* 
Exemplu: .STEP VIN  -.25  .25  .05 ;  .STEP LIN I2 5mA -2mA 0.1mA;

   .STEP  TEMP LIST 0 20 27 50 80 ; .STEP PARAM Cel LIST 0 200u

5. Partea expermimentală

Se vor descrie şi simula PSPICE circuitele discutate în cadrul lucrărilor nr.2 şi 3. În acest sens este necesară parcurgerea următoarelor etape:
a) Se numerotează nodurile schemei electronice. Este de preferat ca potenţialului de referinţă (masa) să i se atribuie valoarea 0, în acest fel toate tensiunile nodurilor vor fi exprimate faţă de masă.

b) Se descrie circuitul prin specificarea tipurilor şi valorilor componentelor şi a legăturilor dintre acestea.
c) Se specifică tipul analizei (c.c., c.a., timp etc). Este permisă rularea unei singure analize la un moment dat.

d) Alte declaraţii – de exemplu cele referitoare la biblioteci de componente sau modelele acestora.

Observaţie: 1. Este recomandat ca un program SPICE să înceapă cu un scurt comentariu (introdus prin caracterul “ * ”) prin care să se descrie conţinutul simulării, numele circuitului sau alte informaţii utile.

2. Un program SPICE se încheie prin directiva .END.

Exemplu. Considerând circuitul din fig.1 să se scrie un program PSPICE destinat studiului timpilor de comutaţie al diodelor.

[image: image3]
Fig.1 Circuit pentru studiul comparativ al timpilor de comutaţie al diodelor

* Studiul comutatiei diodei

*descrierea propriu-zisa a circuitului

Vg 1 0 PULSE(-5V 5V 1n 1n 1n 50n 100n)

R1 1 2 100

R2 1 3 100

D1 2 0 D1N4148

D2 3 0 D1N4001

*specificarea tipului analizei

.TRAN 10n 100n 0 0.1n

*alte declaratii

.LIB DCE.LIB

.PROBE

.END

Fig. 2 Rezultatele simulării PSPICE pentru circuitul din fig.1.

